SECOND FEATURE

THE STEPPER

Step and trace *through 6502 machine language programs with ease! You can set break points, monitor memory locations, and more, with the Stepper!*

The Stepper is a powerful step-and-trace debugger that executes an assembly language program line by line on the Apple II, II Plus, IIe or IIc. It has features that help assembly language programmers track down even the most elusive bugs. It lets you change the contents of the A, X, or Y Registers and the Stack Pointer. You can set break points and reset flags, or scan through several lines in succession.

USING STEPPER

The comments in the program listing (**Listing 1**) explain how the code works. Therefore, the following comments will be directed primarily to the Stepper's features and operation.

BRUNning Stepper enables the Monitor's Control-Y command. To begin stepping, enter from the Monitor the address at which you want to begin, enter Control-Y and press Return. For example,

```
*4000 Control-Y-Return
```

would start the trace at $4000 with all of the Stepper's flags initialized. If for some reason you exit the Stepper, you may start again where you left off by entering **Control-Y-Return**, without adding an address.

On the screen you'll see an inverse prompt line that shows the available options, the current register contents (including the stack pointer and status byte), and the instruction stored in the current address. When you want to proceed, press the Space bar. The displayed instruction will be executed, and another prompt line is displayed. This single-step mode can be continued until a break (BRK) is encountered or until the last RTS empties the stack. Stepper then returns you to the Monitor.

To begin stepping, enter from the Monitor the address at which you want to begin, enter a Control-Y and press Return.

Options

Your options are as follows:

1. To alter the contents of the A, X or Y Registers or the stack pointer, press the A, X, Y or S key. The current contents will be displayed in inverse below the regular display. You may then enter any 8-bit hex value desired.
2. To alter one of the bits in the status byte, press the N, V, D, I, Z or C key. This will toggle the current value and display the new value just below it. For instance, if the Carry bit is zero, pressing C will change it to a one. Lower-case b and e will toggle the B bit and the unused bit, respectively, in the status register. However, the BREAK bit is always set when Stepper is running and the unused bit is always set by the 6502.
3. Pressing Q will exit the Stepper. You may reenter at the current address with the current values by typing **Control-Y-Return**. If you alter a zero-page value at $55 or below, you must enter with an address parameter to allow the Stepper to pick up the current contents of zero page. (See the Zero Page section below.)
4. Pressing T sets/resets the Stack Display flag. When the flag is set, the contents of up to 11 stack locations just below the current stack pointer are displayed after each step, on top of the prompt line.
5. Pressing P sets/resets the Printer flag. When set, output is directed to a printer. The printer format is somewhat different from the screen display. To increase speed, no prompt line is included, and all printing is done in 80 columns instead of 40. If your printer is in a slot other than slot 1, see Modifications to accommodate another slot.

6. Pressing J sets/resets the JSR flag. When set, Stepper will execute all JSR's without stepping through them. Monitor routines, DOS routines, and routines known to be bug-free may be executed without time-consuming single-steps. Stepper will execute the subroutine and return, displaying the next instruction after the JSR.
7. Pressing R cancels the pause after each step. The Stepper traces continuously through your routine until encountering a BRK, a final JSR, or a break point. (See number 8 below.) Pressing any key returns to single-step mode.
8. K and Control-K (denoted in the listings by [K]) are the break-point controls. K allows you to enter a break point. When you are in run mode (R) and an instruction at an address listed among the break points is encountered, the Stepper halts and returns to single-step mode. The break point is not cleared and may be used again. Control-K erases the entire break-point table. Up to five break points may be in effect at one time.
9. E and Control-E ([E]) are the memory display controls. E allows you to enter an address. After each step, Stepper displays this address and its contents above the prompt line. Control-E clears the memory display table. (M would make more sense, but Control-M (alias, Return) could easily be pressed accidentally.)

THE ZERO PAGE

To allow an operation that is as independent as possible, Stepper saves the portion of the zero page it uses and swaps it in and out as needed. On entry into the Stepper when an address parameter is specified, the values in addresses $00-$55 are saved for use by the source program. Entry without an address parameter uses the last-stored zero page. Be careful if any values from $00 to $55 are changed; reenter the Stepper with an address parameter specified. To allow stepping through I/O routines, the four addresses DOS uses to store its input and print data are also swapped by the Stepper.

Stepper cannot step through ProDOS MLI calls, because ProDOS enables alternate banks of memory in the upper 16K, and disconnects RAM in the process. Since Stepper is in RAM, it cannot function properly.

ENTERING THE PROGRAM

If you have an assembler, you may enter the assembly code in **Listing 1** and assemble it to create a working program. If you don't have an assembler, you may enter the hexadecimal code directly from the Monitor. If you are running ProDOS, change HIMEM to protect the code by entering:

```
HIMEM:35584
```

at the Applesoft prompt before you enter the program. Save the program with:

LISTING 1: STEPPER

```
                        1   ********************************
                        2   * STEPPER                      *
                        3   * BY WAYNE EASTWOOD            *
                        4   * COPYRIGHT (C) 1987           *
                        5   * BY MICROSPARC, INC.          *
                        6   * CONCORD, MA  01742           *
                        7   ********************************
                        8   *MERLIN PRO ASSEMBLER
                        9
                       10   * My 0-page use
                       11   CH       =     $24         ;cursor position locations
                       12   CV       =     $25
                       13   LMNEM    =     $2C         ;used in opcode disassembly
                       14   RMNEM    =     $2D
                       15   LENGTH   =     $2F         ;instruction length - 1
                       16   PC       =     $3A         ;PC used by PRADR1
                       17   A1       =     $3C         ;loc. of 16-bit # on entry (MMMM[Y])
                       18
                       19   * Constant definitions
                       20
                       21   CR       =     $8D         ;C/R
                       22   SPACE    =     $A0         ;ASCII space
                       23
                       24   * System and DOS calls
                       25
                       26   STACK    =     $0100       ;6502 stack
                       27   HOOKUP   =     $03EA       ;DOS hook up
                       28   YVECTOR  =     $03F8       ;jump to here on ctrl-Y
                       29   DOSINOUT =     $AA59       ;48K RAM DOS I/O values storage
                       30   KYBOARD  =     $C000       ;key input
                       31   STROBE   =     $C010       ;key strobe
                       32   PRODOS   =     $BF00
                       33   VECTIN   =     $BE32
                       34   VECTOUT  =     $BE30
                       35
                       36   * Monitor calls
                       37
                       38   INSDS2   =     $F88E       ;look up opcode info
                       39   PRADR1   =     $F910       ;print 16-bit address (mid-routine)
                       40   PRBLNX   =     $F948       ;print 3 blanks
                       41   MNEML    =     $F9C0       ;mnemonic table
                       42   MNEMH    =     $FA00
                       43   SETTXT   =     $FB39       ;txt mode & full window
                       44   VTAB     =     $FC22       ;cursor VTAB
                       45   RDKEY    =     $FD0C       ;read key
                       46   CROUT    =     $FD8E       ;output C/R
                       47   PRYX3    =     $FD99       ;print 16-bit at X,Y
                       48   PRBYTE   =     $FDDA       ;print hex in A
                       49   COUT     =     $FDED       ;char out
                       50   SETINV   =     $FE80       ;set screen inverse
                       51   SETNORM  =     $FE84       ;  "     "    normal
                       52   OUTPORT  =     $FE95       ;reset I/O
                       53   MON      =     $FF69       ;reenter monitor
                       54
                       55
                       56   * Assemble one in high memory and one in low memory
                       57   *   to allow tracing programs at various locations
                       58
                       59            ORG   $8F00
                       60
                       61
                       62   * Enters here on BRUN or monitor G
                       63   *   to set up [Y] vector & initialize
                       64   *   program parameters
                       65
8F00: A9 4C            66   ENTER    LDA   #$4C        ;'JMP' is stored for ctrl-Y
8F02: 8D F8 03         67            STA   YVECTOR
8F05: A9 1D            68            LDA   #<STEP      ;point to STEPPER address
8F07: 8D F9 03         69            STA   YVECTOR+1
8F0A: A9 8F            70            LDA   #>STEP
8F0C: 8D FA 03         71            STA   YVECTOR+2
8F0F: A2 0A            72            LDX   #10         ;clear all information registers
8F11: A9 00            73   ENTER10  LDA   #0
8F13: 9D 28 91         74   ENTER20  STA   aSAVE,X
8F16: CA               75            DEX
8F17: 10 FA            76            BPL   ENTER20
8F19: CE 2B 91         77            DEC   sSAVE       ;all new stack
8F1C: 60               78            RTS
                       79
                       80
                       81   * Enter here: ADRS[Y] = step at new PCNT, save current 0-page
```

```
BSAVE STEPPER,A$8F00,L$686
```

If you are using Key Perfect, you should make changes to the object file before you run Key Perfect. BLOAD the file, enter the Monitor with CALL −151 and perform the following instructions:

```
9024:00 00 00
9123:00
9124<9123.9201M
```

Then save the modified program using the BSAVE instruction above.

For help with entering *Nibble* programs, see "A Welcome to New *Nibble* Readers" at the beginning of this issue.

ASSEMBLY

Assemble the Stepper wherever you like. I keep two versions, one at $8F00 and one at $805, so that one will be ready wherever my source program happens to be. I've written the code as compactly as possible while still providing user-friendly operation.

The use of $2C warrants some explanation. In an assembly source listing it looks like the following:

```
        LDA   #0
        HEX   2C
TOGGLE  LDA   #1
```

etc.

but it assembles as:

```
        LDA   #0
        BIT   $01A9
```

etc.

Entering from the top loads the Accumulator with zero. The program then falls through the BIT operation with the Accumulator unscathed. Yet the program may enter at TOGGLE instead to load the Accumulator with a one, saving a byte of code.

At times I've used the ADC operation with the Carry bit set to avoid a CLC, thus saving a byte. In such cases, the result is greater by one than might be expected.

MODIFICATIONS

If your printer interface is not in slot 1, change the AND #1 instructions in **lines 482, 565, 625, 692 and 696** to AND #*n*, where *n* is the slot number of your printer interface. Reassemble the file.

If you are making the changes from the Monitor, BLOAD the object file, enter the Monitor with CALL −151 and make the following changes:

```
92E0:n
938E:n
9418:n
94A9:n
94B4:n
```

BSAVE the resulting file according to the instructions in the Entering the Program section.

LISTING 1: STEPPER *(continued)*

```
                    82  *                [Y] = step at current PCNT, use prev 0-page
                    83
8F1D: 8A            84  STEP    TXA             ;if X=0, then no ADRS
8F1E: F0 12         85          BEQ   STEPLOOP  ;go directly to main routine
8F20: 20 6B 95      86          JSR   ZSAVE0    ;new ADRS, so save current 0-page
8F23: A2 05         87          LDX   #5        ;clear registers & run flag
8F25: 20 11 8F      88          JSR   ENTER10
8F28: A5 3C         89          LDA   A1        ;put new address into PCNT
8F2A: 8D 83 95      90          STA   PCNT      ;A1 set from monitor routine
8F2D: A5 3D         91          LDA   A1+1      ;  when reading [Y]
8F2F: 8D 84 95      92          STA   PCNT+1    ;fall into main tracing routine
                    93
                    94
                    95  * Main tracing routine:
                    96  *    1) Display memory and stack per user direction
                    97  *    2) Display current registers & next instruction
                    98  *    3) Seek user input for step/run, register change, etc.
                    99  *    4) Perform next instruction
                   100  *    5) Return to 1) above
                   101
                   102  *    1) Display memory and stack per user direction
                   103
8F32: 20 39 FB     104  STEPLOOP JSR   SETTXT     ;always start with new screen
8F35: 20 8E FD     105          JSR   CROUT     ;insure new line
8F38: 20 F0 FD     106          JSR   COUT+3    ;2 for screen for readability
8F3B: AE 31 91     107          LDX   mCOUNT    ;display any memory?
8F3E: F0 03        108          BEQ   SL10      ;no
8F40: 20 2C 94     109          JSR   DSPMEM    ;display memory
8F43: AE 30 91     110  SL10    LDX   tFLAG     ;display stack?
8F46: F0 24        111          BEQ   SL40      ;no
8F48: AE 2B 91     112          LDX   sSAVE     ;get current stack location
8F4B: E8           113          INX             ;at bottom?
8F4C: F0 1E        114          BEQ   SL40      ;yes, none to see
8F4E: A0 0A        115          LDY   #10       ;11 max
8F50: BD 00 01     116  SL20    LDA   STACK,X   ;display values separated w/spaces
8F53: 20 DA FD     117          JSR   PRBYTE
8F56: A9 A0        118          LDA   #SPACE
8F58: 20 ED FD     119          JSR   COUT
8F5B: E8           120          INX             ;more stack?
8F5C: F0 03        121          BEQ   SL30      ;no
8F5E: 88           122          DEY             ;more list?
8F5F: 10 EF        123          BPL   SL20      ;yes
8F61: 20 48 F9     124  SL30    JSR   PRBLNX    ;flush with 3 blanks to tab printer
8F64: 88           125          DEY             ;  (not noticed on screen display)
8F65: 10 FA        126          BPL   SL30
8F67: A9 8D        127          LDA   #CR       ;new line for screen but not printer
8F69: 20 F0 FD     128          JSR   COUT+3
                   129
                   130  *    2) Display current registers & next instruction
                   131
8F6C: 20 03 92     132  SL40    JSR   DSPREG    ;display current registers
8F6F: AE 83 95     133          LDX   PCNT      ;print PCNT as hex followed by '-'
8F72: AC 84 95     134          LDY   PCNT+1
8F75: 86 3A        135          STX   PC        ;save for disassembly by PRADR1
8F77: 84 3B        136          STY   PC+1
8F79: 20 99 FD     137          JSR   PRYX3     ;print
8F7C: 20 24 95     138          JSR   ZRESTORE  ;restore source 0-page
8F7F: A2 02        139          LDX   #2        ;get next instruction (3-bytes max)
8F81: 20 82 95     140  SL50    JSR   PRGGET    ;  & store for program reference
8F84: 9D 25 91     141          STA   NXTINST,X ;  at NXTINST, NXTINST+1, NXTINST+2
8F87: CA           142          DEX
8F88: 10 F7        143          BPL   SL50
8F8A: 20 51 95     144          JSR   ZSAVE     ;return my own 0-page
8F8D: A9 EA        145          LDA   #$EA      ;clear execution area with
8F8F: 8D 25 90     146          STA   EXECUTE+1 ;  'NOP' ($EA) instructions
8F92: 8D 26 90     147          STA   EXECUTE+2
8F95: AD 25 91     148          LDA   NXTINST   ;load opcode of instr. & use monitor
8F98: 20 8E F8     149          JSR   INSDS2    ;  to find format, len, & index
8F9B: A2 03        150          LDX   #3        ;three chars in mnemonic
8F9D: A8           151          TAY             ;mnemonic table index
8F9E: B9 C0 F9     152          LDA   MNEML,Y   ;fetch 3-char mnemonic
8FA1: 85 2C        153          STA   LMNEM     ;  (packed in 2 bytes)
8FA3: B9 00 FA     154          LDA   MNEMH,Y   ;(this is an imitation of the Mon.
8FA6: 85 2D        155          STA   RMNEM     ;  rout. except we disass. the
8FA8: A9 00        156  SL60    LDA   #0        ;  opcode only w/o displaying hex
8FAA: A0 05        157          LDY   #5        ;  code as well and mod. the spacing)
8FAC: 06 2D        158  SL70    ASL   RMNEM     ;shift 5 bits of chr into A
8FAE: 26 2C        159          ROL   LMNEM
8FB0: 2A           160          ROL
8FB1: 88           161          DEY
8FB2: D0 F8        162          BNE   SL70
8FB4: 69 BF        163          ADC   #"?       ;carry clear, add '?' offset
8FB6: 20 ED FD     164          JSR   COUT      ;print it
8FB9: C9 BF        165          CMP   #"?       ;exit on unknown opcode '???'
8FBB: D0 0F        166          BNE   SL80
8FBD: 20 80 FE     167          JSR   SETINV    ;display code
8FC0: AD 25 91     168          LDA   NXTINST
8FC3: 20 DA FD     169          JSR   PRBYTE
8FC6: 20 84 FE     170          JSR   SETNORM
8FC9: 4C 40 90     171          JMP   INTBRK10  ;exit
8FCC: CA           172  SL80    DEX
8FCD: D0 D9        173          BNE   SL60
8FCF: A9 A0        174          LDA   #SPACE    ;one blank
8FD1: 20 ED FD     175          JSR   COUT
8FD4: A4 2F        176          LDY   LENGTH    ;prepare to enter norm. Mon. stream
8FD6: A2 06        177          LDX   #6        ;  to print the address field
```

```
8FD8: 20 10 F9  178             JSR   PRADR1      ;enter
                179
                180 *    3) Seek user input for step/run, register change, etc.
                181
8FDB: 20 62 92  182             JSR   RESPOND     ;read user input
                183
                184 *    4) Perform next instruction
                185
8FDE: AE 2B 91  186             LDX   sSAVE       ;ready to go, put on source stack
8FE1: 9A        187             TXS
8FE2: AD 25 91  188             LDA   NXTINST     ;get opcode
                189
                190 * Normally, instr. fall through to PERFORM.  However, we
                191 *   must interpret BRK, RTI, RTS, JSR, JMP, and JMP (XXXX)
                192 *   to maintain control of program.  Relative braching is
                193 *   controlled at PERFORM.
                194
8FE5: F0 56     195             BEQ   INTBRK      ;'BRK'
8FE7: C9 40     196             CMP   #$40        ;'RTI'?
8FE9: F0 58     197             BEQ   INTRTI
8FEB: C9 60     198             CMP   #$60        ;'RTS'?
8FED: F0 5D     199             BEQ   INTRTS
8FEF: C9 20     200             CMP   #$20        ;'JSR'?
8FF1: D0 05     201             BNE   SL90        ;no
8FF3: AE 2E 91  202             LDX   jFLAG       ;yes, run entire routine or step?
8FF6: F0 68     203             BEQ   INTJSR      ;continue single step
8FF8: C9 4C     204 SL90        CMP   #$4C        ;'JMP'?
8FFA: F0 76     205             BEQ   INTJMP
8FFC: C9 6C     206             CMP   #$6C        ;indirect JMP?
8FFE: F0 7D     207             BEQ   INTJMPI
9000: A4 2F     208 PERFORM     LDY   LENGTH      ;get length of instruction
9002: 29 1F     209             AND   #%00011111  ;is it a relative branch opcode?
9004: 49 14     210             EOR   #%00010100
9006: C9 04     211             CMP   #%00000100  ;if equal, branch--have made
9008: F0 03     212             BEQ   SL110       ;  the offset branch into our prgm
900A: B9 25 91  213 SL100       LDA   NXTINST,Y   ;move all necess., others 'NOP'
900D: 99 24 90  214 SL110       STA   EXECUTE,Y
9010: 88        215             DEY
9011: 10 F7     216             BPL   SL100
9013: 20 24 95  217             JSR   ZRESTORE    ;put on source 0-page
9016: AD 2C 91  218             LDA   pSAVE       ;restore all registers
9019: 48        219             PHA
901A: AD 28 91  220             LDA   aSAVE
901D: AE 29 91  221             LDX   xSAVE
9020: AC 2A 91  222             LDY   ySAVE
9023: 28        223             PLP
                224
9024: 00 00 00  225 EXECUTE     DS    3           ;execute the instruction
                226
9027: 4C B1 90  227             JMP   NOBRNCH     ;if no branch, continue
                228
                229
                230 * Branch if taken falls here always.  PCNT is adjusted
                231 *   then we return to STEPLOOP for next instruction
                232
902A: 20 51 95  233             JSR   ZSAVE       ;restore my 0-page
902D: D8        234             CLD               ;restore correct conditions
902E: 18        235             CLC
902F: AD 26 91  236             LDA   NXTINST+1   ;forward or backward branch?
9032: 10 03     237             BPL   XQT10       ;forward
9034: CE 84 95  238             DEC   PCNT+1      ;backward, adjust prgm counter
9037: 20 A5 90  239 XQT10       JSR   ADDPCNT     ;adjust value of prgm counter
903A: 4C C7 90  240             JMP   UPDATE      ;get new prgm counter & step again
                241
                242
                243 * Opcode interpretation routines
                244
903D: 20 03 92  245 INTBRK      JSR   DSPREG      ;display registers
9040: 4C F9 92  246 INTBRK10    JMP   EXIT        ;exit program
                247
9043: BA        248 INTRTI      TSX               ;simulate 'RTI'
9044: E0 FD     249             CPX   #$FD        ;enough on stack?
9046: B0 F8     250             BCS   INTBRK10    ;no
9048: 68        251             PLA               ;pull processor status
9049: 8D 2C 91  252             STA   pSAVE       ;  & continue
904C: BA        253 INTRTS      TSX               ;simulate 'RTS'
904D: E0 FE     254             CPX   #$FE        ;enough on stack?
904F: B0 EF     255             BCS   INTBRK10    ;no
9051: 68        256             PLA               ;pull off return address
9052: 8D 83 95  257             STA   PCNT        ;put on prgm counter
9055: 68        258             PLA
9056: 8D 84 95  259             STA   PCNT+1
9059: BA        260             TSX               ;save stack
905A: 8E 2B 91  261             STX   sSAVE
905D: 4C C7 90  262             JMP   UPDATE      ;next prgm counter & step
                263
9060: AD 83 95  264 INTJSR      LDA   PCNT        ;increment prgm counter by 2
9063: 69 01     265             ADC   #1          ;  imitating normal 6502 operation
9065: A8        266             TAY               ;since carry was set on entry we save
9066: AD 84 95  267             LDA   PCNT+1      ;  SOME CODE BY ADDING 1 plus carry
9069: 69 00     268             ADC   #0          ;  instead of CLC then adding 2
906B: 48        269             PHA               ;and push on stack
906C: 98        270             TYA
906D: 48        271             PHA
906E: BA        272             TSX               ;save & continue as 'JMP'
906F: 8E 2B 91  273             STX   sSAVE
9072: AD 26 91  274 INTJMP      LDA   NXTINST+1   ;find new PC by looking at
9075: 48        275             PHA               ;  opcode address field
```

```
9076: AD 27 91  276             LDA   NXTINST+2
9079: 48        277             PHA
907A: 4C 9A 90  278             JMP   INTJMP10
907D: AD 26 91  279 INTJMPI     LDA   NXTINST+1   ;find indirect address by finding
9080: 8D 83 95  280             STA   PCNT        ;  the contents of the address
9083: AD 27 91  281             LDA   NXTINST+2   ;  specified in the address field
9086: 8D 84 95  282             STA   PCNT+1
9089: 20 24 95  283             JSR   ZRESTORE    ;put back ZP in case adrs is there
908C: A2 00     284             LDX   #0          ;find contents of address
908E: 20 82 95  285             JSR   PRGGET
9091: 48        286             PHA
9092: E8        287             INX
9093: 20 82 95  288             JSR   PRGGET
9096: 48        289             PHA
9097: 20 51 95  290             JSR   ZSAVE       ;swap 0-pages again
909A: 68        291 INTJMP10    PLA               ;install new address
909B: 8D 84 95  292             STA   PCNT+1
909E: 68        293             PLA
909F: 8D 83 95  294             STA   PCNT
90A2: 4C 32 8F  295             JMP   STEPLOOP    ;continue
                296
                297
                298 * Add PCNT to A , carry clear or set prior to entry
                299
90A5: 6D 83 95  300 ADDPCNT     ADC   PCNT        ;if carry set, result 1 greater
90A8: 8D 83 95  301             STA   PCNT
90AB: 90 03     302             BCC   ADDP10
90AD: EE 84 95  303             INC   PCNT+1
90B0: 60        304 ADDP10      RTS
                305
                306
                307 * All non-interpreted instr. that do not branch come here
                308
90B1: 8D 28 91  309 NOBRNCH     STA   aSAVE       ;operation completed, save stuff
90B4: 8E 29 91  310             STX   xSAVE
90B7: 8C 2A 91  311             STY   ySAVE
90BA: 08        312             PHP
90BB: 68        313             PLA
90BC: 8D 2C 91  314             STA   pSAVE
90BF: BA        315             TSX
90C0: 8E 2B 91  316             STX   sSAVE
90C3: 20 51 95  317             JSR   ZSAVE       ;return my 0-page
90C6: D8        318             CLD               ;will cause many problems if set!
90C7: 38        319 UPDATE      SEC               ;add LENGTH + 1 (SEC) to PC
90C8: A5 2F     320             LDA   LENGTH
90CA: 20 A5 90  321             JSR   ADDPCNT     ;add them
90CD: 4C 32 8F  322             JMP   STEPLOOP    ;continue
                323
                324
                325 * Possible user responses
                326
90D0: C1 D8 D9  327 REGNAM      ASC   "AXYS"
90D3: D3
90D4: CE D6 E5  328 PROCNAM     HEX   CED6E5E2C4C9DAC3 ;NVebDICZ (lower-case not used)
90D7: E2 C4 C9 DA C3
90DC: D2 CA D0  329 PSUDNAM     ASC   "RJPTEK"
90DF: D4 C5 CB
90E2: 85 8B     330             HEX   858B        ;ctrl-E, ctrl-K
90E4: 20 03 1A  331 PROMPT      INV   " CZIDB.VN  S  Y  X  A   ]K[K ]E[E R JPTQ"
90E7: 09 04 02 2E 16 0E 20 20
90EF: 13 20 20 19 20 20 18 20
90F7: 20 01 20 20 20 1D 0B 1B
90FF: 0B 20 1D 05 1B 05 20 12
9107: 20 0A 10 14 11
                332
                333 * Cursor tab for user responses
                334
910C: 13 16 19  335 REGTAB      HEX   1316191C
910F: 1C
9110: 1F 20 21  336 PROCTAB     HEX   1F20212223242526
9113: 22 23 24 25 26
9118: 03 02 01  337 PSDTAB      HEX   030201
                338
                339 * Masks for pSAVE
                340
911B: 80 40 20  341 ORMASK      HEX   8040201008040201
911E: 10 08 04 02 01
                342
9123: 00        343 INDEX       DS    1           ;temporary index for user entry
9124: 00        344 DFLAG       DS    1           ;multi-purpose flag
9125: 00 00 00  345 NXTINST     DS    3           ;holds next instruction
9128: 00        346 aSAVE       DS    1           ;register save
9129: 00        347 xSAVE       DS    1
912A: 00        348 ySAVE       DS    1
912B: 00        349 sSAVE       DS    1
912C: 00        350 pSAVE       DS    1
912D: 00        351 rFLAG       DS    1           ;run flag
912E: 00        352 jFLAG       DS    1           ;JSR flag
912F: 00        353 pFLAG       DS    1           ;printer flag
9130: 00        354 tFLAG       DS    1           ;stack flag
9131: 00        355 mCOUNT      DS    1           ;number of memory loc. (mCOUNT+1)/2
9132: 00        356 bCOUNT      DS    1           ;  "   "  brk. points (bCOUNT+1)/2
                357
9133: 00 00 00  358 MEMAREA     DS    2*10-1      ;space for 10 memory locations
9136: 00 00 00 00 00 00 00 00
913E: 00 00 00 00 00 00 00 00
9146: 00 00 00  359 BRKAREA     DS    2*5-1       ;space for 5 breakpoints
```

```
9149: 00 00 00 00 00 00
                        360
914F: 00 00 00   361  PRGZ     DS    $56         ;source program 0-page storage
9152: 00 00 00 00 00 00 00 00
915A: 00 00 00 00 00 00 00 00
9162: 00 00 00 00 00 00 00 00
916A: 00 00 00 00 00 00 00 00
9172: 00 00 00 00 00 00 00 00
917A: 00 00 00 00 00 00 00 00
9182: 00 00 00 00 00 00 00 00
918A: 00 00 00 00 00 00 00 00
9192: 00 00 00 00 00 00 00 00
919A: 00 00 00 00 00 00 00 00
91A2: 00 00 00
91A5: 00 00 00   362  MYZ      DS    $56         ;my 0-page storage
91A8: 00 00 00 00 00 00 00 00
91B0: 00 00 00 00 00 00 00 00
91B8: 00 00 00 00 00 00 00 00
91C0: 00 00 00 00 00 00 00 00
91C8: 00 00 00 00 00 00 00 00
91D0: 00 00 00 00 00 00 00 00
91D8: 00 00 00 00 00 00 00 00
91E0: 00 00 00 00 00 00 00 00
91E8: 00 00 00 00 00 00 00 00
91F0: 00 00 00 00 00 00 00 00
91F8: 00 00 00
91FB: 00 00 00   363  PRGDOS   DS    4           ;program DOS print parameters
91FE: 00
91FF: 00 00 00   364  MYDOS    DS    4           ;my DOS print parameters
9202: 00
                 365
                 366
                 367  * Display current registers
                 368
9203: A2 27      369  DSPREG   LDX   #39         ;prompt for screen only
9205: BD E4 90   370  DSPR10   LDA   PROMPT,X
9208: 20 F0 FD   371          JSR   COUT+3
920B: CA         372          DEX
920C: 10 F7      373          BPL   DSPR10
920E: A9 BE      374          LDA   #">
9210: 20 F0 FD   375          JSR   COUT+3
9213: A2 02      376          LDX   #2            ;examine flags
9215: BD 2E 91   377  DSPR20   LDA   jFLAG,X
9218: F0 03      378          BEQ   DSPR21
921A: 20 80 FE   379          JSR   SETINV        ;display in inverse if set
921D: BD DD 90   380  DSPR21   LDA   PSUDNAM+1,X
9220: 20 F0 FD   381          JSR   COUT+3        ;display flag letter
9223: 20 84 FE   382          JSR   SETNORM
9226: CA         383          DEX
9227: 10 EC      384          BPL   DSPR20
9229: A9 13      385          LDA   #19           ;tab right on screen
922B: 85 24      386          STA   CH
922D: A2 00      387          LDX   #0
922F: BD 28 91   388  DSPR30   LDA   aSAVE,X      ;get value of registers
9232: 20 DA FD   389          JSR   PRBYTE        ;display
9235: A9 A0      390          LDA   #SPACE
9237: 20 ED FD   391          JSR   COUT
923A: E8         392          INX
923B: E0 04      393          CPX   #4            ;done?
923D: 90 F0      394          BCC   DSPR30        ;no
923F: AD 2C 91   395          LDA   pSAVE         ;now display processor status
9242: A8         396          TAY
9243: A9 80      397          LDA   #%10000000    ;mask, interested in bits here
9245: 85 3C      398          STA   A1            ;temporary storage
9247: 98         399  DSPR40   TYA               ;restore pSAVE value
9248: 25 3C      400          AND   A1            ;and w/ current mask
924A: F0 03      401          BEQ   DSPR50        ;0 value
924C: A9 B1      402          LDA   #"1           ;on, display '1'
924E: 2C         403          HEX   2C            ;'BIT XXXX' -- falls through
924F: A9 B0      404  DSPR50   LDA   #"0         ;off, display '0'
9251: 20 ED FD   405          JSR   COUT
9254: 46 3C      406          LSR   A1            ;shift mask -->
9256: D0 EF      407          BNE   DSPR40        ;do for all bits
9258: A9 8D      408          LDA   #CR           ;for screen
925A: 20 F0 FD   409          JSR   COUT+3
925D: A9 BE      410          LDA   #">           ;next line (mark for printer)
925F: 4C ED FD   411          JMP   COUT
                 412
                 413
                 414  * Stop loop and get user input
                 415
9262: AD 2D 91   416  RESPOND  LDA   rFLAG         ;in run mode?
9265: F0 12      417          BEQ   RS10          ;no
9267: 20 01 95   418          JSR   BRKCHK        ;yes, check for breakpoint
926A: B0 08      419          BCS   RS05          ;found one so stop
926C: AD 00 C0   420          LDA   KYBOARD       ;in run mode, any key stops
926F: 10 53      421          BPL   RS70          ;else exit w/out checking further
9271: 8D 10 C0   422          STA   STROBE        ;clear
9274: A9 00      423  RS05     LDA   #0            ;no more run
9276: 8D 2D 91   424          STA   rFLAG
9279: A9 17      425  RS10     LDA   #23           ;position cursor
927B: 85 25      426          STA   CV
927D: A9 11      427          LDA   #17
927F: 85 24      428          STA   CH
9281: 20 22 FC   429          JSR   VTAB
9284: 20 0C FD   430          JSR   RDKEY         ;get response from user
9287: C9 A0      431          CMP   #SPACE        ;next step?
9289: F0 39      432          BEQ   RS70          ;yes, exit
928B: C9 D1      433          CMP   #"Q           ;quit?
928D: D0 03      434          BNE   RS20
928F: 4C F7 92   435          JMP   EXIT0         ;exit one level into JSR's
                 436
                 437  * Check if a register response A,X,Y, or S
                 438
9292: A0 03      439  RS20     LDY   #3
9294: D9 D0 90   440  RS30     CMP   REGNAM,Y
9297: F0 26      441          BEQ   RS60R         ;yes
9299: 88         442          DEY
929A: 10 F8      443          BPL   RS30
                 444
                 445  * Check if one of the processor status bits N,V,D,I,Z, or C
                 446
929C: A0 07      447          LDY   #7
929E: D9 D4 90   448  RS40     CMP   PROCNAM,Y
92A1: F0 16      449          BEQ   RS60P         ;yes
92A3: 88         450          DEY
92A4: 10 F8      451          BPL   RS40
                 452
                 453  * Check if one of the pseudo ops R,E,[E],K,[K],T,P, or J
                 454
92A6: A0 07      455          LDY   #7
92A8: D9 DC 90   456  RS50     CMP   PSUDNAM,Y
92AB: F0 05      457          BEQ   RS60S         ;yes
92AD: 88         458          DEY
92AE: 10 F8      459          BPL   RS50
92B0: 30 C7      460          BMI   RS10          ;no correct response
                 461
                 462  * Handle input: carry set if to proceed to next instruction
                 463
92B2: 20 59 93   464  RS60S    JSR   DOPSUEDO      ;handle pseudo ops
92B5: 90 C2      465          BCC   RS10          ;back for more (not 'R' or step)
92B7: B0 0B      466          BCS   RS70          ;continue to next step
92B9: 20 1D 93   467  RS60P    JSR   DOPROC        ;processor bits
92BC: 4C 79 92   468          JMP   RS10          ;always back for more
92BF: 20 C5 92   469  RS60R    JSR   DOREG         ;register bytes
92C2: 90 B5      470          BCC   RS10          ;no step
92C4: 60         471  RS70     RTS
                 472
92C5: 8C 23 91   473  DOREG    STY   INDEX         ;alter register, Y tells which one
92C8: 20 80 FE   474  R10      JSR   SETINV        ;for emphasis
92CB: 20 94 93   475          JSR   ONSCREEN      ;insure screen only
92CE: AC 23 91   476          LDY   INDEX         ;restore index
92D1: B9 0C 91   477          LDA   REGTAB,Y      ;get proper tab for display
92D4: 85 24      478          STA   CH
92D6: B9 28 91   479          LDA   aSAVE,Y       ;display current value
92D9: 20 DA FD   480          JSR   PRBYTE
92DC: AD 2F 91   481          LDA   pFLAG         ;return to specified output device
92DF: 29 01      482          AND   #1            ;1 or 0
92E1: 20 96 93   483          JSR   DFLTOUT       ;restore current output
92E4: C6 24      484          DEC   CH            ;put cursor on top of image
92E6: 20 84 FE   485          JSR   SETNORM
92E9: 20 0C FD   486          JSR   RDKEY         ;change it as desired
92EC: C9 A0      487          CMP   #SPACE        ;step?
92EE: D0 01      488          BNE   R30           ;no
92F0: 60         489  R20      RTS               ;carry set if space
92F1: C9 D1      490  R30      CMP   #"Q           ;quit?
92F3: D0 0A      491          BNE   R40           ;no
92F5: 68         492          PLA                 ;exit 2 deep in JSR's
92F6: 68         493          PLA
92F7: 68         494  EXIT0    PLA
92F8: 68         495          PLA
92F9: 20 24 95   496  EXIT     JSR   ZRESTORE      ;restore 0-page
92FC: 4C 69 FF   497          JMP   MON           ;exit to monitor
                 498
92FF: 49 B0      499  R40      EOR   #"0           ;determines if hex digit
9301: C9 0A      500          CMP   #$0A          ;digits 0-9 result in value 0-9 so
9303: 90 06      501          BCC   R50           ;  carry clear means yes
9305: 69 88      502          ADC   #$88          ;digits $A-$F to $71-$76, add
9307: C9 FA      503          CMP   #$FA          ;  $89 (w/carry) to push to $FA-$FF
9309: 90 E5      504          BCC   R20           ;no, exit for more input
930B: A0 03      505  R50      LDY   #3            ;lower nibble of A holds $x0-$xF
930D: 0A         506          ASL                 ;move low nibble to high
930E: 0A         507          ASL
930F: 0A         508          ASL
9310: 0A         509          ASL
9311: AE 23 91   510          LDX   INDEX         ;where does this go?
9314: 0A         511  R60      ASL               ;move into low nibble of storage
9315: 3E 28 91   512          ROL   aSAVE,X       ;  & move low nibble to high
9318: 88         513          DEY
9319: 10 F9      514          BPL   R60
931B: 30 AB      515          BMI   R10           ;always
                 516
931D: 8C 23 91   517  DOPROC   STY   INDEX         ;toggle stat bit, Y points to bit
9320: 20 80 FE   518  P10      JSR   SETINV
9323: AC 23 91   519          LDY   INDEX         ;position cursor
9326: B9 10 91   520          LDA   PROCTAB,Y
9329: 85 24      521          STA   CH
932B: AD 2C 91   522          LDA   pSAVE         ;load entire status & mask
932E: 39 1B 91   523          AND   ORMASK,Y      ;  for desired bit
9331: F0 03      524          BEQ   P20           ;it's a 0, convert to 1
9333: A9 B0      525          LDA   #"0           ;it's a 1, convert to 0
9335: 2C         526          HEX   2C            ;'BIT XXXX' falls through 3 bytes
9336: A9 B1      527  P20      LDA   #"1
9338: 20 F0 FD   528          JSR   COUT+3        ;display on screen only
```

```
933B: 20 84 FE  529           JSR   SETNORM
933E: AC 23 91  530           LDY   INDEX
9341: C9 30     531           CMP   #"0-$80     ;update status bit (MSB
9343: F0 08     532           BEQ   P30         ;  cleared due to SETINV)
9345: B9 1B 91  533           LDA   ORMASK,Y    ;change 0 to 1
9348: 0D 2C 91  534           ORA   pSAVE
934B: D0 08     535           BNE   P40         ;always
934D: B9 1B 91  536 P30       LDA   ORMASK,Y
9350: 49 FF     537           EOR   #$FF        ;convert for ANDing
9352: 2D 2C 91  538           AND   pSAVE       ;clear a bit
9355: 8D 2C 91  539 P40       STA   pSAVE
9358: 60        540           RTS
                541
9359: 98        542 DOPSUEDO  TYA               ;handle all other codes
935A: D0 05     543           BNE   PS1         ;not run flag
935C: EE 2D 91  544           INC   rFLAG       ;'R', so set flag
935F: 38        545           SEC               ;run on exit
9360: 60        546           RTS
9361: C9 04     547 PS1       CMP   #4          ;J, P, or T flags?
9363: B0 4D     548           BCS   PSD1        ;no, some sort of display option
9365: AA        549           TAX               ;index for correct flag
9366: A9 FF     550           LDA   #$FF        ;toggle flag
9368: 5D 2D 91  551           EOR   rFLAG,X     ;X=1, 2, or 3
936B: 9D 2D 91  552           STA   rFLAG,X
936E: F0 03     553           BEQ   PSF1        ;now update display, 0=off
9370: 20 80 FE  554           JSR   SETINV      ;on shown in inverse
9373: C6 25     555 PSF1      DEC   CV          ;move cursor
9375: 20 22 FC  556           JSR   VTAB
9378: BD 17 91  557           LDA   PSDTAB-1,X  ;get cursor tab (X is 1 too large)
937B: 85 24     558           STA   CH
937D: BD DC 90  559           LDA   PSUDNAM,X   ;get display letter
9380: 20 F0 FD  560           JSR   COUT+3      ;screen only
9383: 20 84 FE  561           JSR   SETNORM
9386: E0 02     562           CPX   #2          ;printer toggle requested?
9388: D0 26     563           BNE   PSF4        ;no
938A: AD 2F 91  564           LDA   pFLAG       ;yes, get current value
938D: 29 01     565           AND   #1          ;make a 1 or 0
938F: D0 05     566           BNE   DFLTOUT     ;turn printer on
9391: 20 8E FD  567           JSR   CROUT       ;rtn to screen, clr any prtr buff
9394: A9 00     568 ONSCREEN  LDA   #0          ;PR#0
9396: 20 95 FE  569 DFLTOUT   JSR   OUTPORT     ;set up hooks
9399: AD 00 BF  570           LDA   PRODOS
939C: C9 4C     571           CMP   #$4C
939E: D0 0D     572           BNE   HOOKDOS
93A0: A5 36     573           LDA   $36
93A2: 8D 30 BE  574           STA   VECTOUT
93A5: A5 37     575           LDA   $37
93A7: 8D 31 BE  576           STA   VECTOUT+1
93AA: 4C B0 93  577           JMP   PSF4
93AD: 20 EA 03  578 HOOKDOS   JSR   HOOKUP      ;tell DOS
93B0: 18        579 PSF4      CLC               ;return for more user input
93B1: 60        580           RTS
                581
93B2: C9 07     582 PSD1      CMP   #7          ;handle display options, ctrl-K?
93B4: D0 07     583           BNE   PSD2        ;no
93B6: A9 00     584           LDA   #0          ;clear all breakpoints
93B8: 8D 32 91  585           STA   bCOUNT
93BB: F0 F3     586           BEQ   PSF4        ;always
93BD: C9 06     587 PSD2      CMP   #6          ;clear memory (ctrl-E)?
93BF: D0 07     588           BNE   PSD3        ;no
93C1: A9 00     589           LDA   #0          ;clear all
93C3: 8D 31 91  590           STA   mCOUNT
93C6: F0 E8     591           BEQ   PSF4        ;always
93C8: C9 05     592 PSD3      CMP   #5          ;set breakpoints (K)?
93CA: F0 03     593           BEQ   BSET        ;yes
93CC: 4C 23 94  594           JMP   INPMEM      ;display addresses & add to list
93CF: 20 94 93  595 BSET      JSR   ONSCREEN    ;set up for screen only
93D2: 20 8E FD  596           JSR   CROUT
93D5: AE 32 91  597           LDX   bCOUNT      ;bCOUNT=# of entries*2+1
93D8: F0 16     598           BEQ   BS20        ;no entries/no display
93DA: CA        599           DEX
93DB: BD 46 91  600 BS10      LDA   BRKAREA,X   ;stored low/high
93DE: 20 DA FD  601           JSR   PRBYTE      ;display
93E1: CA        602           DEX
93E2: BD 46 91  603           LDA   BRKAREA,X
93E5: 20 DA FD  604           JSR   PRBYTE
93E8: A9 A0     605           LDA   #SPACE      ;separate
93EA: 20 ED FD  606           JSR   COUT
93ED: CA        607           DEX
93EE: 10 EB     608           BPL   BS10
93F0: 20 8E FD  609 BS20      JSR   CROUT       ;done with display
93F3: AE 32 91  610           LDX   bCOUNT      ;full?
93F6: E0 0A     611           CPX   #10
93F8: B0 1A     612           BCS   BS30        ;yes, can't add any
93FA: 20 B8 94  613           JSR   GETDIG      ;get address
93FD: AE 32 91  614           LDX   bCOUNT
9400: AD 24 91  615           LDA   DFLAG       ;any entry?
9403: F0 0F     616           BEQ   BS30        ;no
9405: A5 3C     617           LDA   A1          ;put new address in memory
9407: 9D 46 91  618           STA   BRKAREA,X
940A: E8        619           INX
940B: A5 3D     620           LDA   A1+1
940D: 9D 46 91  621           STA   BRKAREA,X
9410: E8        622           INX
9411: 8E 32 91  623           STX   bCOUNT
9414: AD 2F 91  624 BS30      LDA   pFLAG       ;restore output device
9417: 29 01     625           AND   #1
9419: 20 96 93  626           JSR   DFLTOUT
941C: 68        627           PLA               ;pull off JSR's
941D: 68        628           PLA
941E: 68        629           PLA
941F: 68        630           PLA
9420: 4C 32 8F  631           JMP   STEPLOOP    ;redo current line
                632
                633
                634 * Display routines: INPMEM = display then add
                635 *                  DSPMEM = display only
                636
9423: 20 94 93  637 INPMEM    JSR   ONSCREEN    ;set up for screen only
9426: 20 8E FD  638           JSR   CROUT
9429: A9 00     639           LDA   #0          ;put a 0 as input flag
942B: 2C        640           HEX   2C          ;'BIT XXXX' 3-byte fall through
942C: A9 BD     641 DSPMEM    LDA   #"=         ;store and use as display flag
942E: 8D 24 91  642           STA   DFLAG
9431: AE 31 91  643           LDX   mCOUNT      ;mCOUNT=# of entries*2+1
9434: F0 55     644           BEQ   DM50        ;no entries/no display
9436: CA        645           DEX
9437: A9 09     646           LDA   #9          ;10*8 across for printer
9439: 8D 23 91  647           STA   INDEX       ;makes printer 80 column before C/R
943C: BD 33 91  648 DM10      LDA   MEMAREA,X   ;stored low/high
943F: 8D 5C 94  649           STA   DMXX+2      ;for contents display
9442: 20 DA FD  650           JSR   PRBYTE
9445: CA        651           DEX
9446: BD 33 91  652           LDA   MEMAREA,X
9449: 8D 5B 94  653           STA   DMXX+1
944C: 20 DA FD  654           JSR   PRBYTE
944F: AD 24 91  655           LDA   DFLAG       ;0=address only
9452: F0 11     656           BEQ   DM20        ;input req. (do not disp contents)
9454: 20 ED FD  657           JSR   COUT        ;print flag ('=')
9457: 20 24 95  658           JSR   ZRESTORE    ;pull in proper 0-page
945A: AD FF FF  659 DMXX      LDA   $FFFF       ;read memory contents (avoid
945D: 48        660           PHA               ;  0-page conflict)
945E: 20 51 95  661           JSR   ZSAVE       ;give me back my 0-page
9461: 68        662           PLA               ;restore memory contents
9462: 20 DA FD  663           JSR   PRBYTE      ;print it
9465: CE 23 91  664 DM20      DEC   INDEX       ;keep up with line len for printer
9468: 10 0A     665           BPL   DM30
946A: 20 8E FD  666           JSR   CROUT       ;new line for printer
946D: A9 09     667           LDA   #9          ;new count for line length
946F: 8D 23 91  668           STA   INDEX
9472: D0 05     669           BNE   DM40        ;always
9474: A9 A0     670 DM30      LDA   #SPACE      ;separate entries w/space
9476: 20 ED FD  671           JSR   COUT
9479: CA        672 DM40      DEX               ;next entry
947A: 10 C0     673           BPL   DM10
947C: 20 8E FD  674           JSR   CROUT       ;done
947F: AD 24 91  675           LDA   DFLAG       ;input required?
9482: D0 2C     676           BNE   DM70        ;no
9484: AE 31 91  677           LDX   mCOUNT      ;full?
9487: E0 14     678           CPX   #20
9489: B0 1A     679           BCS   DM60        ;yes
948B: 20 B8 94  680 DM50      JSR   GETDIG      ;get new address, rouse DFLAG
948E: AE 31 91  681           LDX   mCOUNT
9491: AD 24 91  682           LDA   DFLAG       ;any entry?
9494: F0 0F     683           BEQ   DM60        ;no
9496: A5 3C     684           LDA   A1          ;put new address in memory
9498: 9D 33 91  685           STA   MEMAREA,X
949B: A5 3D     686           LDA   A1+1
949D: E8        687           INX
949E: 9D 33 91  688           STA   MEMAREA,X
94A1: E8        689           INX
94A2: 8E 31 91  690           STX   mCOUNT
94A5: AD 2F 91  691 DM60      LDA   pFLAG       ;restore output device
94A8: 29 01     692           AND   #1
94AA: 20 96 93  693           JSR   DFLTOUT
94AD: 4C 32 8F  694           JMP   STEPLOOP    ;redo current line
94B0: AD 2F 91  695 DM70      LDA   pFLAG       ;restore output device
94B3: 29 01     696           AND   #1
94B5: 4C 96 93  697           JMP   DFLTOUT     ;& return to tracing stream
                698
                699
                700 * Get 16-bit digit
                701
94B8: A9 00     702 GETDIG    LDA   #0          ;clear workspace
94BA: 85 3C     703           STA   A1
94BC: 85 3D     704           STA   A1+1
94BE: 8D 24 91  705           STA   DFLAG       ;set only on a valid entry
94C1: A9 17     706 GD10      LDA   #23         ;position cursor
94C3: 85 25     707           STA   CV
94C5: A9 11     708           LDA   #17
94C7: 85 24     709           STA   CH
94C9: 20 22 FC  710           JSR   VTAB
94CC: 20 80 FE  711           JSR   SETINV      ;for emphasis
94CF: A5 3D     712           LDA   A1+1        ;print current value
94D1: 20 DA FD  713           JSR   PRBYTE      ;to scrn only (due to calling rout)
94D4: A5 3C     714           LDA   A1
94D6: 20 DA FD  715           JSR   PRBYTE
94D9: C6 24     716           DEC   CH          ;put cursor on image
94DB: 20 84 FE  717           JSR   SETNORM
94DE: 20 0C FD  718           JSR   RDKEY       ;input
94E1: 49 B0     719           EOR   #"0         ;hex digit? (see R40 above)
94E3: C9 0A     720           CMP   #$0A
94E5: 90 06     721           BCC   GD20
```

```
94E7: 69 88      722           ADC   #$88
94E9: C9 FA      723           CMP   #$FA
94EB: 90 13      724           BCC   GD40        ;no
94ED: EE 24 91   725  GD20     INC   DFLAG       ;set for valid entry (if <> 256!)
94F0: A2 03      726           LDX   #3
94F2: 0A         727           ASL               ;shift to high nibble
94F3: 0A         728           ASL
94F4: 0A         729           ASL
94F5: 0A         730           ASL
94F6: 0A         731  GD30     ASL
94F7: 26 3C      732           ROL   A1
94F9: 26 3D      733           ROL   A1+1        ;move carry into storage
94FB: CA         734           DEX
94FC: 10 F8      735           BPL   GD30
94FE: 30 C1      736           BMI   GD10        ;more digits
9500: 60         737  GD40     RTS
                 738
                 739
                 740  * Check current PCNT against breakpoints
                 741
9501: AE 32 91   742  BRKCHK   LDX   bCOUNT      ;any to check?
9504: F0 15      743           BEQ   BC30        ;no
9506: CA         744           DEX
9507: BD 46 91   745  BC10     LDA   BRKAREA,X   ;stored low/high
950A: CA         746           DEX
950B: CD 84 95   747           CMP   PCNT+1      ;current location
950E: D0 08      748           BNE   BC20        ;no match
9510: BD 46 91   749           LDA   BRKAREA,X
9513: CD 83 95   750           CMP   PCNT
9516: F0 05      751           BEQ   BC40        ;match!
9518: CA         752  BC20     DEX               ;keep looking
9519: 10 EC      753           BPL   BC10
951B: 18         754  BC30     CLC               ;no match
951C: 60         755           RTS
951D: A9 87      756  BC40     LDA   #$87        ;bell
951F: 20 ED FD   757           JSR   COUT
9522: 38         758           SEC
9523: 60         759           RTS
                 760
                 761
                 762  * Zero-page movers: I only use $55 and below
                 763  *   (Also DOS I/O storage so I/O stuff can be traced)
                 764
                 765
9524: A0 55      766  ZRESTORE LDY   #$55        ;save local, restore source
9526: B9 00 00   767  ZR1      LDA   0,Y
9529: 99 A5 91   768           STA   MYZ,Y
952C: 88         769           DEY
952D: 10 F7      770           BPL   ZR1
952F: A0 03      771           LDY   #3          ;same for DOS memory
9531: B9 59 AA   772  ZR2      LDA   DOSINOUT,Y
9534: 99 FF 91   773           STA   MYDOS,Y
9537: 88         774           DEY
9538: 10 F7      775           BPL   ZR2
953A: A0 55      776           LDY   #$55
953C: B9 4F 91   777  ZR3      LDA   PRGZ,Y
953F: 99 00 00   778           STA   0,Y
9542: 88         779           DEY
9543: 10 F7      780           BPL   ZR3
9545: A0 03      781           LDY   #3
9547: B9 FB 91   782  ZR4      LDA   PRGDOS,Y
954A: 99 59 AA   783           STA   DOSINOUT,Y
954D: 88         784           DEY
954E: 10 F7      785           BPL   ZR4
9550: 60         786           RTS
                 787
9551: 20 6B 95   788  ZSAVE    JSR   ZSAVE0      ;save source, restore local
9554: A0 55      789           LDY   #$55
9556: B9 A5 91   790  ZS3      LDA   MYZ,Y
9559: 99 00 00   791           STA   0,Y
955C: 88         792           DEY
955D: 10 F7      793           BPL   ZS3
955F: A0 03      794           LDY   #3
9561: B9 FF 91   795  ZS4      LDA   MYDOS,Y
9564: 99 59 AA   796           STA   DOSINOUT,Y
9567: 88         797           DEY
9568: 10 F7      798           BPL   ZS4
956A: 60         799           RTS
                 800
956B: A0 55      801  ZSAVE0   LDY   #$55        ;save source
956D: B9 00 00   802  ZS1      LDA   0,Y
9570: 99 4F 91   803           STA   PRGZ,Y
9573: 88         804           DEY
9574: D0 F7      805           BNE   ZS1
9576: A0 03      806           LDY   #3
9578: B9 59 AA   807  ZS2      LDA   DOSINOUT,Y
957B: 99 FB 91   808           STA   PRGDOS,Y
957E: 88         809           DEY
957F: 10 F7      810           BPL   ZS2
9581: 60         811           RTS
                 812
                 813
                 814  * Program counter and opcode reader
                 815
9582: BD FF FF   816  PRGGET   LDA   $FFFF,X     ;avoid 0 page conflict
9585: 60         817           RTS
                 818
                 819  PCNT     =     PRGGET+1    ;simulated program counter
END OF LISTING 1
```

KEY PERFECT 5.0
RUN ON
STEPPER

CODE-5.0	ADDR# - ADDR#	CODE-4.0
73BCF4DE	8F00 - 8F4F	264B
23BF025F	8F50 - 8F9F	26DE
5B8431DD	8FA0 - 8FEF	2B45
CF55E927	8FF0 - 903F	23A4
8C5B12E0	9040 - 908F	27C1
00722474	9090 - 90DF	2A36
8869FF3E	90E0 - 912F	2AF9
5678BE35	9130 - 917F	00
5678BE35	9180 - 91CF	00
036A32FB	91D0 - 921F	0E4E
5F41BD90	9220 - 926F	2957
33D9BD61	9270 - 92BF	299D
E389FDB4	92C0 - 930F	24C8
A1921B2B	9310 - 935F	2AFE
EA9926F6	9360 - 93AF	256E
C59FACA3	93B0 - 93FF	2A2B
0068F924	9400 - 944F	27E2
A6E23A44	9450 - 949F	2925
8CDBDBE4	94A0 - 94EF	2569
8DA29921	94F0 - 953F	225A
B95B6853	9540 - 9585	2265
57ADFA97 = PROGRAM TOTAL =		0686